
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.3.1.21 which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2015-03_D15wk25’ ATS delivery.

The test case can be demonstrated to run with one LTE-A UE (see section 5) on multiple test platforms. Execution logs are provided as evidence.
2. Table of Contents

21. Overview

2. Table of Contents
2
3. Verification Test Summary
2
4. Corrections required for TC 8.3.1.21
3
Change 1 – Correction to function ‘f_TC_8_3_1_21_EUTRA’
3
5. Execution Log Files
7
5.1 Qualcomm MDM 8984
7
References
7

3. Verification Test Summary

Test Case:
TC_8_3_1_21
Test Group:
LTE_A\8_3\ EUTRA_Measurements_eICIC.ttcn
ATS Version:
iwd-TTCN3-B2015-03_D15wk25
System Simulator used:
R&S CMW500 + Anritsu Protocol Conformance Test System ME7832L
UE used:
Qualcomm MDM 8994
Verification Status:
PASS


4. Corrections required for TC 8.3.1.21
Change 1 – Correction to function ‘f_TC_8_3_1_21_EUTRA’ 
	Function name
	f_TC_8_3_1_21_EUTRA () 

	Reason for change
	1.    Cell 2 has to be “on” and power level changes in step 3 have to be moved before GetNextSendOccasion is true on cell 2, as GetNextSendOccasion would not work for an “off” cell.

2.    Event A4 criteria may also be met after the handover, which would result in the UE continuing to send Measurement Reports after step 4.

	Summary of change
	1.    f_EUTRA_SetCellPowerList moved before f_EUTRA_GetNextSendOccasion in step 3 with changing to “timing_now”.

2.    Measurement configuration removed via modified Reconfiguration Request message contents in step 5. 
Note: This change requires a Prose CR.

	TTCN module
	\LTE_A\8_3\EUTRA_Measurements_eICIC.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted conditional to prose CR agreement. Implemented using the template cs_ReportConfigToRemoveList_1Entry


Before change

...

  //================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.3.1.21
   * @status    APPROVED (LTE_A)
   */
  function f_TC_8_3_1_21_EUTRA() runs on EUTRA_PTC
  { /* eICIC / Measurement configuration control and reporting / Event A4 Handover / Neighbour RSRP and RSRQ measurement configuration change */
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var PhysCellId v_PhysCellId_Cell2;
    var Frequency_IE_Type v_Frequency_IE;
    var ChannelBandwidthDependency_Type v_ChBandDependency;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo;
    var template (value) MeasSubframePattern_r10 v_MeasSubframePatternNeigh_r10;
    var template (value) MeasSubframePatternConfigNeigh_r10 v_MeasSubframePatternConfigNeigh_r10;
    var SubFrameTiming_Type v_TimingInfo;
    var template(value) CellPowerList_Type v_CellPowerListT1:= {
      cs_CellPower(eutra_Cell2, -91)// @sic R5-151752 sic@
    };
    //Initialise variables
    f_EUTRA_Init(c1);
    v_FDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);
    v_Frequency_IE := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency := f_EUTRA_BandDependentParam(v_Frequency_IE.DL_ChBandwidth, v_Frequency_IE.UL_ChBandwidth);
    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    //Cell 1 and Cell 2 are time & SFN synchronous cells
    f_EUTRA_InitialiseWithSameCellTiming ({eutra_Cell1, eutra_Cell2});
    //Create and configure Cell1, Cell2
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    //Preamble: The UE is in state Generic RB Established (state 3) on Cell 1
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_RbEst_Def(eutra_Cell1);
    f_EUTRA_TestBody_Set(true);
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1-2" siclog@
    //@sic R5-151752 sic@
    //The SS transmits an RRCConnectionReconfiguration message including measConfig and neighbour cell measurement restriction pattern on Cell 1 to setup intra frequency measurement reporting for event A4.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1 to confirm the setup of intra frequency measurement.
    if (v_FDD_TDDInfo.FDD_TDD == FDD) {
      v_MeasSubframePatternNeigh_r10 := {
        subframePatternFDD_r10 := '1000000010000000100000001000000010000000'B //@sic R5-151752 sic@
      };
    } else {
      v_MeasSubframePatternNeigh_r10 := {
        subframePatternTDD_r10 := {
          subframeConfig1_5_r10:= '00001000010000100001'B //@sic R5-151752 sic@
        }
      };
    }
    v_MeasSubframePatternConfigNeigh_r10 := {
      setup := {
        measSubframePatternNeigh_r10 := v_MeasSubframePatternNeigh_r10,
        measSubframeCellList_r10 := {
          {
            start_ := v_PhysCellId_Cell2,
            range := omit
          }
        }
      }
    }
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,
                                                                                                      cs_MeasConfig(omit,
                                                                                                                    cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                                                                                                     cs_MeasObject_measObjectEUTRA_Common(v_Frequency_IE.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency.AllowedMeasBandwidth, -, -, -, v_MeasSubframePatternConfigNeigh_r10)),
                                                                                                                    omit,
                                                                                                                    cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A4,//@sic R5-151752 sic@
                                                                                                                                                       cs_ReportConfig_reportConfigEUTRA_eventA4_RSRQ(-10, 0, ms320, rsrq, both, 1, ms1024, infinity_)),
                                                                                                                    omit,
                                                                                                                    cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A4))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 2");
    //@siclog "Step 3" siclog@
    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in Table 8.3.1.21.3.2-1.
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(eutra_Cell2);
    f_EUTRA_SetCellPowerList(v_CellPowerListT1, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number));
    //OFDMA Channel Noise Generator (OCNG) is required to populate Cell 2. The Cell will be populated in all sub-frames specified in measSubframePatternNeigh as '0' and will be removed where defined as '1'.
    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_PhysicalLayerConfigDL_REQ(eutra_Cell2, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), cs_PhysicalLayerConfigDL_PDSCH_RelativeTxPower_Ocng));
    f_EUTRA_ActivateOCNG ( eutra_Cell2, v_MeasSubframePatternNeigh_r10, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number));
    //@siclog "Step 4" siclog@
    //@sic R5-151752 sic@
    //Check: Does the UE transmit a MeasurementReport message on Cell 1 to report event A4 with the measured RSRP and RSRQ values for Cell 2?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_Eutra_1Entry_both(1, v_PhysCellId_Cell2)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 4");
    //@siclog "Step 5-6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to order the UE to perform intra frequency handover to Cell 2.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 using common preamble to confirm the successful completion of the intra frequency handover?
    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1, eutra_Cell2 );
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 6");
    //End TestBody
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_DeactivateOCNG (eutra_Cell2);
    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);
  } //end of testcase 8.3.1.21
... 
After change

... 

  //================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.3.1.21
   * @status    APPROVED (LTE_A)
   */
  function f_TC_8_3_1_21_EUTRA() runs on EUTRA_PTC
  { /* eICIC / Measurement configuration control and reporting / Event A4 Handover / Neighbour RSRP and RSRQ measurement configuration change */
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var PhysCellId v_PhysCellId_Cell2;
    var Frequency_IE_Type v_Frequency_IE;
    var ChannelBandwidthDependency_Type v_ChBandDependency;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo;
    var template (value) MeasSubframePattern_r10 v_MeasSubframePatternNeigh_r10;
    var template (value) MeasSubframePatternConfigNeigh_r10 v_MeasSubframePatternConfigNeigh_r10;
    var SubFrameTiming_Type v_TimingInfo;
    var template(value) CellPowerList_Type v_CellPowerListT1:= {
      cs_CellPower(eutra_Cell2, -91)// @sic R5-151752 sic@
    };
    //Initialise variables
    f_EUTRA_Init(c1);
    v_FDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);
    v_Frequency_IE := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency := f_EUTRA_BandDependentParam(v_Frequency_IE.DL_ChBandwidth, v_Frequency_IE.UL_ChBandwidth);
    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    //Cell 1 and Cell 2 are time & SFN synchronous cells
    f_EUTRA_InitialiseWithSameCellTiming ({eutra_Cell1, eutra_Cell2});
    //Create and configure Cell1, Cell2
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    //Preamble: The UE is in state Generic RB Established (state 3) on Cell 1
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_RbEst_Def(eutra_Cell1);
    f_EUTRA_TestBody_Set(true);
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1-2" siclog@
    //@sic R5-151752 sic@
    //The SS transmits an RRCConnectionReconfiguration message including measConfig and neighbour cell measurement restriction pattern on Cell 1 to setup intra frequency measurement reporting for event A4.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1 to confirm the setup of intra frequency measurement.
    if (v_FDD_TDDInfo.FDD_TDD == FDD) {
      v_MeasSubframePatternNeigh_r10 := {
        subframePatternFDD_r10 := '1000000010000000100000001000000010000000'B //@sic R5-151752 sic@
      };
    } else {
      v_MeasSubframePatternNeigh_r10 := {
        subframePatternTDD_r10 := {
          subframeConfig1_5_r10:= '00001000010000100001'B //@sic R5-151752 sic@
        }
      };
    }
    v_MeasSubframePatternConfigNeigh_r10 := {
      setup := {
        measSubframePatternNeigh_r10 := v_MeasSubframePatternNeigh_r10,
        measSubframeCellList_r10 := {
          {
            start_ := v_PhysCellId_Cell2,
            range := omit
          }
        }
      }
    }
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,
                                                                                                      cs_MeasConfig(omit,
                                                                                                                    cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                                                                                                     cs_MeasObject_measObjectEUTRA_Common(v_Frequency_IE.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency.AllowedMeasBandwidth, -, -, -, v_MeasSubframePatternConfigNeigh_r10)),
                                                                                                                    omit,
                                                                                                                    cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A4,//@sic R5-151752 sic@
                                                                                                                                                       cs_ReportConfig_reportConfigEUTRA_eventA4_RSRQ(-10, 0, ms320, rsrq, both, 1, ms1024, infinity_)),
                                                                                                                    omit,
                                                                                                                    cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A4))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 2");
    //@siclog "Step 3" siclog@
    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in Table 8.3.1.21.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerListT1);
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(eutra_Cell2);
    //OFDMA Channel Noise Generator (OCNG) is required to populate Cell 2. The Cell will be populated in all sub-frames specified in measSubframePatternNeigh as '0' and will be removed where defined as '1'.
    f_EUTRA_SS_CommonCellConfig(eutra_Cell2, cas_PhysicalLayerConfigDL_REQ(eutra_Cell2, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), cs_PhysicalLayerConfigDL_PDSCH_RelativeTxPower_Ocng));
    f_EUTRA_ActivateOCNG ( eutra_Cell2, v_MeasSubframePatternNeigh_r10, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number));
    //@siclog "Step 4" siclog@
    //@sic R5-151752 sic@
    //Check: Does the UE transmit a MeasurementReport message on Cell 1 to report event A4 with the measured RSRP and RSRQ values for Cell 2?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_Eutra_1Entry_both(1, v_PhysCellId_Cell2)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 4");
    //@siclog "Step 5-6" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to order the UE to perform intra frequency handover to Cell 2.
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 using common preamble to confirm the successful completion of the intra frequency handover?
    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1, eutra_Cell2, -, -, cs_MeasConfig(cs_MeasObjectToRemoveList_1Entry(1), omit, cs_ReportConfigToRemovedList_1Entry(6), omit, cs_MeasIdToRemoveList_1Entry(1), omit));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.21 Step 6");
    //End TestBody
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_DeactivateOCNG (eutra_Cell2);
    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);
  } //end of testcase 8.3.1.21

... 

MCC160 Additional changes 
Change 1 – test case selection expression
 According to the draft prose CR on 36.523-2, the selection expression C154F and C154T shall be consistent:

  function f_SelectionExpr ( Condition_Type p_Condition) return boolean {

  /* Set the applicability conditions according to 36.523-2 Table 4-1a: Applicability of tests Conditions */

    var boolean v_ApplCond;

    select (p_Condition) {

      //R: UEs supporting E-UTRA

      case (R) {

        v_ApplCond := true;

      }

      //C01: UEs supporting E-UTRA and UTRA

…
      //C154: UEs supporting E-UTRA and Feature Group Indictor 115

      case (C154){

        if ( fl_IsPrimaryBandFdd() ) {        //@sic R5-150557 sic@

            v_ApplCond := pc_eFDD and pc_FeatrGrp_115_F;

        } else {

            v_ApplCond := pc_eTDD and pc_FeatrGrp_115_T;

        }

      }

…

Change 2 - code improvement 

Some optimisation for EUTRA_CellInfo are added as improvement: mainly avoiding inline templates in f_EUTRA_InitSystemInformation in order to reduce the processing time by 2-3 times.
f_EUTRA_InitSystemInformation

Similar changes are done in 

· f_EUTRA_ChangeSysinfoCombinationC1_ToC21

· f_EUTRA_ChangeSysinfoCombinationC21_ToC1

· f_EUTRA_ChangeSIB5_Combination3

	  function f_EUTRA_InitSystemInformation(EUTRA_CellId_Type                  p_CellId,

                                         CellIdentity                       p_CellIdentity,

                                         Combination_Type                   p_Combination,

                                         FreqBandIndicator                  p_FrequencyBand,

                                         Dl_Bandwidth_Type                  p_DL_ChBandwidth,

                                         Ul_Bandwidth_Type                  p_UL_ChBandwidth,

                                         template (value) PLMN_IdentityList p_PLMN_Identity,

                                         TrackingAreaCode                   p_TrackingAreaCode,

                                         EUTRA_FDD_TDD_CellInfo_Type        p_FDD_TDD,

                                         RootSequenceIndex_Type             p_RootSequenceIndex)

    return template (value) BcchInfo_Type

  { /* @sic R5-096641: parameterise RootSequenceIndex sic@ */

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(p_DL_ChBandwidth, p_UL_ChBandwidth);

    var EUTRA_SysinfoDL_Bandwidth_Type v_SysinfoDL_Bandwidth := fl_EUTRA_ConvertDL_Bandwidth_ToSysinfoType (p_DL_ChBandwidth);

    var template(omit) TDD_Config v_TDD_Config := omit;  // omitted in FDD

    var boolean v_QbasedTestCase := f_GetTestcaseAttrib_Qbased_Rsrq(testcasename());  // Set to TRUE if QBASED test case  @sic R5-121384 sic@

    var template (value) BcchInfo_Type v_CellBcchInfo;

    var template (value) BCCH_BCH_Message v_MIB;

    var template (value) BCCH_DL_SCH_Message v_SIB1; 
    var template (value) SI_List_Type v_SI_List;
    var template (value) SystemInformation_r8_IEs v_SIB2;

    var template (value) SystemInformation_r8_IEs v_SIB3;

    var template (value) SystemInformation_r8_IEs v_SIB5;

    var template (value) SystemInformation_r8_IEs v_SIB6;

    var template (value) SystemInformation_r8_IEs v_SIB7;

    var template (value) SystemInformation_r8_IEs v_SIB8;

    var template (value) SystemInformation_r8_IEs v_TwoSIBs;

    var template (value) SystemInformationBlockType5 v_SystemInformation5;

    var template (value) SoundingRS_UL_ConfigCommon v_SoundingRS_UL_ConfigCommon;

    var template (value) PRACH_ConfigSIB v_PRACH_ConfigSIB;

    // --- Initialise MIB ---

    v_MIB :=  cs_508_MasterInformationBlock_Init(v_SysinfoDL_Bandwidth);

    // --- Initialise variables for SIB1 ---

    if (p_FDD_TDD.FDD_TDD == TDD) {  // initialise tdd-Configuration  to be used in SIB1

      v_TDD_Config := cs_TDD_ConfigDef(p_FDD_TDD.TDD_SubframeAssignment);    // Value omitted in FDD

    }

    // --- Initialise SIB1 ---

    // -- A default scheduling assigned ----------

    if (v_QbasedTestCase) {  // @sic R5-121384 sic@

      v_SIB1 := cds_508_SystemInformationBlockType1_QBased(p_PLMN_Identity,

                                                           p_TrackingAreaCode,

                                                           p_CellIdentity,

                                                           p_FrequencyBand,

                                                           cs_SchedulingInfo_2SI( v_ChannelBandwidthDependency.SIB3_periodicity),

                                                           v_TDD_Config);

    } else {

      v_SIB1 := cs_508_SystemInformationBlockType1_Def(p_PLMN_Identity,

                                                       p_TrackingAreaCode,

                                                       p_CellIdentity,

                                                       p_FrequencyBand,

                                                       cs_SchedulingInfo_2SI( v_ChannelBandwidthDependency.SIB3_periodicity),

                                                       v_TDD_Config);

    }

    // --- Initialise SIB2 ---

    if (p_FDD_TDD.FDD_TDD == FDD) {            // FDD

      v_SoundingRS_UL_ConfigCommon := cs_508_SoundingRS_UL_ConfigCommon_FDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration);

      v_PRACH_ConfigSIB := cs_508_PRACH_ConfigSIB_FDD(v_ChannelBandwidthDependency.Prach_FrequencyOffset, p_RootSequenceIndex);

    } else {                                   // TDD

      v_SoundingRS_UL_ConfigCommon := cs_508_SoundingRS_UL_ConfigCommon_TDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration);

      v_PRACH_ConfigSIB := cs_508_PRACH_ConfigSIB_TDD(v_ChannelBandwidthDependency.Prach_FrequencyOffset, p_RootSequenceIndex);

    }

    v_SIB2 := cs_SI_SIB2(cs_508_SystemInformationBlockType2_Def(omit,

                                                                v_ChannelBandwidthDependency.N_RB_CQI,

                                                                v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                v_SoundingRS_UL_ConfigCommon,

                                                                v_PRACH_ConfigSIB,

                                                                tsc_P_b_1Tx));

    // --- Initialise SIB3 ---

    if (v_QbasedTestCase) {  // @sic R5-121384 sic@

      v_SIB3 := cs_SI_SIB3 (cds_508_SystemInformationBlockType3_Qbased);

    } else {

      v_SIB3 := cs_SI_SIB3 (cs_508_SystemInformationBlockType3_Def);

    }

    // --- Initialise all SYSINFO --

    select (p_Combination) {

      case (c1) {

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList :=  cs_SchedulingInfo_2SI( v_ChannelBandwidthDependency.SIB3_periodicity);
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2), 

          cs_SystemInformation_Common(v_SIB3) 

        };
        v_CellBcchInfo := cs_BCCHInfo_C1(v_MIB,

                                         v_SIB1,

                                         { cs_SystemInformation_Common(v_SIB2), cs_SystemInformation_Common(v_SIB3) });

      }

      case (c2) { // SIB4 Intra-Frequency

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList :=  cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                    {sibType3},

                                                                                                    v_ChannelBandwidthDependency.SIB4_periodicity,

                                                                                                    {sibType4});
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)) 

        };
        v_CellBcchInfo := cs_BCCHInfo_C1(v_MIB,

                                         v_SIB1,

                                         { cs_SystemInformation_Common(v_SIB2),

                                           cs_SystemInformation_Common(v_SIB3),

                                           cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)) }

                                         );

      }

      case (c3) { // SIB5 Inter-Frequency

        v_SystemInformation5 := f_EUTRA_InitSystemInformation5 (p_CellId , v_ChannelBandwidthDependency.AllowedMeasBandwidth, p_DL_ChBandwidth, p_UL_ChBandwidth, p_FrequencyBand);

        v_SIB5 := cs_SI_SIB5(v_SystemInformation5);

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList :=  cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                    {sibType3},

                                                                                                    v_ChannelBandwidthDependency.SIB5_periodicity,

                                                                                                    {sibType5}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB5)

        };






      }

      case (c4) { // SIB6 UTRAN

        v_SIB6 := cs_SI_SIB6( cs_508_SystemInformationBlockType6_Def(omit, omit));  // Initialise first UTRA FDD and TDD frequency list with OMIT

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB6_periodicity,

                                                                                                   {sibType6}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB6)

        };






      }

      case (c5) {   // SIB7 GERAN

        v_SIB7 := cs_SI_SIB7( cs_508_SystemInformationBlockType7_Def(omit));  // List of GERAN frequencies assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB7_periodicity,

                                                                                                   {sibType7}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB7)

        };






      }

      case (c6) { // SIB2/SIB3/SIB8 for CDMA2000 @sic R5-103870 sic@

        v_SIB8 := cs_SI_SIB8( cs_508_SystemInformationBlockType8_Def(omit, omit));  // 1XRTT or HRPD parameters assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB8_periodicity,

                                                                                                   {sibType8}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB8)

        };






      }

      case (c7) { // for eNodeB: broadcast SIB2/SIB3/SIB4/SIB9

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_4SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB4_periodicity,

                                                                                                   {sibType4},

                                                                                                   v_ChannelBandwidthDependency.SIB9_periodicity,

                                                                                                   {sibType9}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)),

          cs_SystemInformation_Common(cs_SI_SIB9(cs_SystemInformationBlockType9_Def(char2oct ("3gppTest"))))

        };







      }

      case (c8) { // for ETWS: broadcast SIB2/SIB3/SIB10/SIB11

        // Test cases using c8 shall be initialised with C1, C8 is started after the preamble

        FatalError (__FILE__, __LINE__, "invalid sysinfo combination c8 shall be started inside the test case");

      }

      case (c9) {  // SIB2/SIB3/SIB5 Inter-Frequency + SIB6 UTRAN

        v_SystemInformation5 := f_EUTRA_InitSystemInformation5 (p_CellId , v_ChannelBandwidthDependency.AllowedMeasBandwidth,p_DL_ChBandwidth, p_UL_ChBandwidth, p_FrequencyBand);

        v_TwoSIBs := cs_SI_SIB5_SIB6( v_SystemInformation5,

                                      cs_508_SystemInformationBlockType6_Def(omit, omit)); // Initialise first UTRA FDD and TDD frequency list with OMIT

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB5_periodicity,   // SIB5 and SIB6 have the same periodicity

                                                                                                   {sibType5, sibType6});  // SIB5 and SIB6 in the same SI 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_TwoSIBs)

        };






      }

      case (c10) { //  SIB2/SIB3/SIB6 UTRAN + SIB7 GERAN

        v_SIB6 := cs_SI_SIB6( cs_508_SystemInformationBlockType6_Def(omit, omit));  // Initialise first UTRA FDD and TDD frequency list with OMIT

        v_SIB7 := cs_SI_SIB7( cs_508_SystemInformationBlockType7_Def(omit));  // List of GERAN frequencies assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_4SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB6_periodicity,

                                                                                                   {sibType6},

                                                                                                   v_ChannelBandwidthDependency.SIB7_periodicity,

                                                                                                   {sibType7}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB6),

          cs_SystemInformation_Common(v_SIB7)

        };







      }

      case (c10a) { //  SIB2/SIB3/SIB5/SIB6 UTRAN + SIB7 GERAN

        v_SystemInformation5 := f_EUTRA_InitSystemInformation5(p_CellId , v_ChannelBandwidthDependency.AllowedMeasBandwidth,p_DL_ChBandwidth, p_UL_ChBandwidth, p_FrequencyBand);

        v_TwoSIBs := cs_SI_SIB5_SIB6(v_SystemInformation5,

                                     cs_508_SystemInformationBlockType6_Def(omit, omit)); // Initialise first UTRA FDD and TDD frequency list with OMIT

        v_SIB7 := cs_SI_SIB7(cs_508_SystemInformationBlockType7_Def(omit));  // List of GERAN frequencies assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_4SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB5_periodicity,   // SIB5 and SIB6 have the same periodicity

                                                                                                   {sibType5, sibType6},

                                                                                                   v_ChannelBandwidthDependency.SIB7_periodicity,

                                                                                                   {sibType7}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_TwoSIBs),

          cs_SystemInformation_Common(v_SIB7)

        };







      }

      case (c11) { // SIB2 + SIB3 + SIB4 + SIB6 UTRAN + SIB9

        v_SIB6 := cs_SI_SIB6( cs_508_SystemInformationBlockType6_Def(omit, omit));  // Initialise first UTRA FDD and TDD frequency list with OMIT

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_5SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB4_periodicity,

                                                                                                   {sibType4},

                                                                                                   v_ChannelBandwidthDependency.SIB6_periodicity,

                                                                                                   {sibType6},

                                                                                                   v_ChannelBandwidthDependency.SIB9_periodicity,

                                                                                                   {sibType9}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)),

          cs_SystemInformation_Common(v_SIB6),

          cs_SystemInformation_Common(cs_SI_SIB9(cs_SystemInformationBlockType9_Def(char2oct ("3gppTest"))))

        };








      }

      case (c12) { // SIB2 + SIB3 + SIB5 + SIB7 GERAN   @sic R5-106551 sic@

        v_SystemInformation5 := f_EUTRA_InitSystemInformation5(p_CellId, v_ChannelBandwidthDependency.AllowedMeasBandwidth, p_DL_ChBandwidth, p_UL_ChBandwidth, p_FrequencyBand);

        v_SIB5 := cs_SI_SIB5(v_SystemInformation5);

        v_SIB7 := cs_SI_SIB7(cs_508_SystemInformationBlockType7_Def(omit));  // List of GERAN frequencies assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_4SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB5_periodicity,

                                                                                                   {sibType5},

                                                                                                   v_ChannelBandwidthDependency.SIB7_periodicity,

                                                                                                   {sibType7}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(v_SIB5),

          cs_SystemInformation_Common(v_SIB7)

        };







      }

      case (c13) { // SIB2 + SIB3 + SIB4 + SIB5 + SIB9 @sic R5-106551 R5-110471 sic@

        v_SystemInformation5 := f_EUTRA_InitSystemInformation5(p_CellId , v_ChannelBandwidthDependency.AllowedMeasBandwidth, p_DL_ChBandwidth, p_UL_ChBandwidth, p_FrequencyBand);

        v_SIB5 := cs_SI_SIB5(v_SystemInformation5);

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_5SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB4_periodicity,

                                                                                                   {sibType4},

                                                                                                   v_ChannelBandwidthDependency.SIB5_periodicity,

                                                                                                   {sibType5},

                                                                                                   v_ChannelBandwidthDependency.SIB9_periodicity,

                                                                                                   {sibType9}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)),

          cs_SystemInformation_Common(v_SIB5),

          cs_SystemInformation_Common(cs_SI_SIB9(cs_SystemInformationBlockType9_Def(char2oct ("3gppTest"))))

        };








      }

      case (c14) { // SIB2 + SIB3 + SIB4 + SIB7 + SIB9 @sic R5-110471 sic@

        v_SIB7 := cs_SI_SIB7(cs_508_SystemInformationBlockType7_Def(omit));  // List of GERAN frequencies assigned later after synchronisation with GERAN PTC

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList := cs_SchedulingInfo_5SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                   {sibType3},

                                                                                                   v_ChannelBandwidthDependency.SIB4_periodicity,

                                                                                                   {sibType4},

                                                                                                   v_ChannelBandwidthDependency.SIB7_periodicity,

                                                                                                   {sibType7},

                                                                                                   v_ChannelBandwidthDependency.SIB9_periodicity,

                                                                                                   {sibType9}); 
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB4( cs_508_SystemInformationBlockType4_Def)),

          cs_SystemInformation_Common(v_SIB7),

          cs_SystemInformation_Common(cs_SI_SIB9(cs_SystemInformationBlockType9_Def(char2oct ("3gppTest"))))

        };








      }

      case (c15, c16, c18, c19, c20) {

        // for MBMS

        // Test cases using c15/c16/c18/c19/c20 shall be initialised with C1. c15/c16/c18/c19/c20 are initialised after the preamble

        FatalError(__FILE__, __LINE__, "invalid sysinfo combination c15/c16/c18/c19/c20 shall be started inside the test case");

      }

      case (c17) { // for CMAS: broadcast SIB2/SIB3/SIB12

                   // Test cases using c17 shall be initialised with C1, C17 is started after the preamble

        FatalError(__FILE__, __LINE__, "invalid sysinfo combination c17 shall be started inside the test case");

      }

      case (c21) { // for SIMTC - SIB2/SIB3/SIB14 @sic R5-140075 sic@

        v_SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList :=  cs_SchedulingInfo_3SI(v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                                    {sibType3},

                                                                                                    v_ChannelBandwidthDependency.SIB14_periodicity,

                                                                                                    {sibType14_v1130});
        v_SI_List := { 

          cs_SystemInformation_Common(v_SIB2),

          cs_SystemInformation_Common(v_SIB3),

          cs_SystemInformation_Common(cs_SI_SIB14( cs_SystemInformationBlockType14_Def))

        };






      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid sysinfo combination");

      }

    } 
    v_CellBcchInfo := cs_BCCHInfo_C1(v_MIB, v_SIB1, v_SI_List);
    return v_CellBcchInfo;

  }


5. Execution Log Files

5.1 Qualcomm MDM 8984
The Qualcomm MDM 8994 UE passed this LTE-A eICIC test case on R&S CMW500 Multi-RAT Protocol Tester and on Anritsu Protocol Conformance Test System ME7832L in eFDD band 3. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Rohde-Schwarz\tc_8_3_1_21_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

· Test Case Execution log file: 
Anritsu\static_TC_8_3_1_21@2015-07-10_12.54.09_PASS_FDD   
(Note: PICS/PIXIT settings are captured in file “ModuleParameters_FDD.txt”)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

